

## Assignment 1 Question 2

Copyright, Michael Monagan, September 2023

```
> Maple2Array := proc(f::polynom,x::name)
  local d,A,i;
  if f=0 then return Array(0..-1); fi; # the empty array
  d := degree(f,x);
  A := Array(0..d);
  for i from 0 to d do A[i] := coeff(f,x,i); od;
  A;
end:

Array2Maple := proc(A::Array,x::name)
  local r,d,i;
  r := [op(2,A)]; # range(s) for subscripts
  if nops(r)<>1 or lhs(r[1])<>0 then
    error "Array must be one dimensional and indexed from 0"
  fi;
  d := rhs(r[1]); # upper index
  add( A[i]*x^i, i=0..d );
end:

> DEGREE := proc(A::Array)
  local r;
  r := [op(2,A)];
  if nops(r)<>1 or lhs(r[1])<>0 then
    error "Array must be one dimensional and indexed from 0"
  fi;
  rhs(r[1]);
end:

COEFF := proc(A::Array,n::nonnegint)
  local d;
  d := DEGREE(A);
  if n>d then 0 else A[n] fi;
end:

DIFF := proc(A::Array)
  local d,D,i;
  d := DEGREE(A);
  if d<1 then return Array(0..-1) fi; # return 0
  D := Array(0..d-1);
  for i to d do D[i-1] := i*A[i] od;
  D;
end:

EVAL := proc(A::Array,z::rational)
  local d,y,i;
  d := DEGREE(A);
```

```

    if d<0 then return 0 fi;
    y := A[d];
    for i from d-1 by -1 to 0 do
        y := z*y+A[i];
    od;
    y;
end:

```

```

> f := 3*x^3-5*x+6;
A := Maple2Array(f,x);
DEGREE(A);

```

$$f := 3x^3 - 5x + 6$$

$$A := [6, -5, 0, 3, \dots 0 \dots 3 \text{ Array}]$$

(1)

```

> COEFF(A,3);
COEFF(A,6);

```

$$3$$

$$0$$

(2)

```

> EVAL(A,2);
eval(f,x=2); # check

```

$$20$$

$$20$$

(3)

```

> DIFF(A);
DEGREE(DIFF(A));
diff(f,x); # check
Array2Maple(DIFF(A),x);

```

$$[-5, 0, 9, \dots 0 \dots 2 \text{ Array}]$$

$$2$$

$$9x^2 - 5$$

$$9x^2 - 5$$

(4)